```
// The case m = 2
F := Rationals();
PR<t> := PolynomialRing(F);
F<w> := NumberField(t^2 + t + 1);
F<u> := FunctionField(F);
c := u^3;

F<t> := FunctionField(F);
E2 := EllipticCurve([0, t^2 + c]);
P := [];
P[1] := E2 ! [-u, t, 1];
P[2] := E2 ! [-w*u, t, 1];

M := HeightPairingMatrix(P);
M;
Determinant(M);



////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////



// The case m = 3
F<v> := QuadraticField(-3);
w := (-1 + v)/2;
F<u> := FunctionField(F);
c := -u^6/108;
F<t> := FunctionField(F);
E3 := EllipticCurve([0, t^3 + c]);
P := [];

x := -t;
y := v*u^3/18;
P[1] := E3 ! [x, y, 1];
P[3] := E3 ! [w*x, y, 1];

x := -t + u^2/3;
y := u*t - u^3/6;
P[2] := E3 ! [x, y, 1];
P[4] := E3 ! [w*x, y, 1];

M := HeightPairingMatrix(P);
M;
Determinant(M);
```

```
////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////



// The case m = 4
F := Rationals();
PR<t> := PolynomialRing(F);
F<i,w> := NumberField([t^2 + 1, t^2 + t + 1]);
s3 := (2*w + 1)/i;
assert(s3^2 eq 3);
v := 26*s3 - 45;
assert( (2*s3 - 3)^3 eq 3*v );

F<u> := FunctionField(F);
c := u^12/(2^6*3*v);
cc := (2*s3 + 3)*u^4/12;
assert(cc^3 eq c);

F<t> := FunctionField(F);
E4 := EllipticCurve([0, t^4 + c]);
P := [];

x := -cc;
y := t^2;
P[1] := E4 ! [x, y, 1];
P[4] := E4 ! [w*x, y, 1];

x := u*t + (s3 + 3)*u^4/12;
y := t^2 + u^3*t/2 + (s3 + 2)*u^6/8;
P[2] := E4 ! [x, y, 1];
P[5] := E4 ! [w*x, y, 1];

x := i*u*t + (s3 + 3)*u^4/12;
y := t^2 - i*u^3*t/2 - (s3 + 2)*u^6/8;
P[3] := E4 ! [x, y, 1];
P[6] := E4 ! [w*x, y, 1];

M := HeightPairingMatrix(P);
M;
Determinant(M);



////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
```

```
// The case m = 5
// Unlike m < 5, Magma free online calculator is not so powerful to completely
// determine at once the Gram matrix of the height pairing on the points Pk.
// Nevertheless, the matrix can be found by launching the function “HeightPairing”
// with u = 1, c = 1/(60*theta) and separately for a few pairs of the points.
// It is worth noting that the height does not depend on the basic field, hence
// the matrix remains unchanged regardless of c. Indeed, there is a simple
// isomorphism (over a certain field extension) between the surfaces E5 with the
// two given coefficients c.

F := Rationals();
PR<t> := PolynomialRing(F);
F<w,z> := NumberField([ t^2 + t + 1, t^4 + t^3 + t^2 + t + 1 ]);
s5 := 2*z^3 + 2*z^2 + 1;
assert(s5^2 eq 5);
v := z^2*(z - 1)*(2*w + 1);
assert( v^2 eq 3*(s5 + 5)/2 );
theta := 564300 + 252495*s5 + 170252*v + 76074*s5*v;

F<u0> := FunctionField(F);
c := u0^30/(60*theta);
// u0 := 1;
// c := 1/(60*theta);
F<t> := FunctionField(F);
E5 := EllipticCurve([0, t^5 + c]);

P := [];
for i := 0 to 3 do
for j := 0 to 1 do
    u := w^j*z^i*u0;
    a0 := -( (8289*z^3 + 35113*z^2 + 43402*z + 21701)*w +
    (26238*z^3 + 39650*z^2 + 21701*z - 2804) )*u^10/15;
    a1 := -( (58*z^3 + 246*z^2 + 304*z + 152)*w +
    (184*z^3 + 278*z^2 + 152*z - 19) )*u^4/5;
    a2 := 1/u^2;

    b0 := ( 12*a0*a1 - a1^3*u^2 - 12*a0*u^4 + 15*a1^2*u^6 + 9*a1*u^10 + u^14 )*u/16;
    b1 := ( 12*a0 + 3*a1^2*u^2 - 6*a1*u^6 - u^10 )/(8*u);
    b2 := ( 3*a1 + u^4 )/(2*u);
    b3 := 1/u^3;

    x := a2*t^2 + a1*t + a0;
    y := b3*t^3 + b2*t^2 + b1*t + b0;
    k := 4*j + i + 1;
    P[k] := E5 ! [x, y, 1];
```

```
    // Height(P[k]);
end for;
end for;

/*
HeightPairing(P[1], P[2]); // -1
HeightPairing(P[1], P[3]); // 0
HeightPairing(P[1], P[4]); // 0
HeightPairing(P[1], P[5]); // -1
HeightPairing(P[1], P[6]); // 1
HeightPairing(P[1], P[7]); // 0
HeightPairing(P[1], P[8]); // 0
*/
/*
HeightPairing(P[2], P[3]); // -1
HeightPairing(P[2], P[4]); // 0
HeightPairing(P[2], P[5]); // 0
HeightPairing(P[2], P[6]); // -1
HeightPairing(P[2], P[7]); // 1
HeightPairing(P[2], P[8]); // 0
*/
/*
HeightPairing(P[3], P[4]); // -1
HeightPairing(P[3], P[5]); // 0
HeightPairing(P[3], P[6]); // 0
HeightPairing(P[3], P[7]); // -1
HeightPairing(P[3], P[8]); // 1
*/
/*
HeightPairing(P[4], P[5]); // 0
HeightPairing(P[4], P[6]); // 0
HeightPairing(P[4], P[7]); // 0
HeightPairing(P[4], P[8]); // -1
*/
/*
HeightPairing(P[5], P[6]); // -1
HeightPairing(P[5], P[7]); // 0
HeightPairing(P[5], P[8]); // 0
HeightPairing(P[6], P[7]); // -1
HeightPairing(P[6], P[8]); // 0
HeightPairing(P[7], P[8]); // -1
*/
```

//

```
F := Rationals();
M := Matrix(F, 8, 8,
    [2, -1,  0,  0, -1,  1,  0,  0,
    -1,  2, -1,  0,  0, -1,  1,  0,
     0, -1,  2, -1,  0,  0, -1,  1,
     0,  0, -1,  2,  0,  0,  0, -1,
    -1,  0,  0,  0,  2, -1,  0,  0,
     1, -1,  0,  0, -1,  2, -1,  0,
     0,  1, -1,  0,  0, -1,  2, -1,
     0,  0,  1, -1,  0,  0, -1,  2]);
Determinant(M);
```